
Abstract. 在巴西，不動產估價被視為屬於工程領域。根據 ABNT NBR 14653-1 標
準，房地產估價師（engenheiro de avaliações）被定義為經專業培訓且在 CREA 註冊
的大學畢業生；而 CONFEA 第 345/1990 號決議則賦予工程專業人員進行檢查、實
施專家鑑定、執行估價以及裁決有爭議之技術問題的職責（arbitramento）。房地產經
紀人（corretor de imóveis）依據第 6.530/1978 號法律（Lei 6.530/1978）獨立設立，
具備技術資格，其法定職權為仲介及就房地產交易提供意見；至於此職權是否延伸至
估價領域，兩大專業理事會間曾就此展開訴訟。本文探討這兩項角色的區別、確立此
區分的監管措施，以及所涉及的相應責任。文中檢視了 CNAI 的經紀人-估價師名冊，
並分析了兩個專業理事會之間的分歧。本文結論指出，巴西雖已對估價流程進行規
範，卻未為價值確立穩固的基礎，並將此方法與德國的「雙價值體系」進行比較。附
錄以 Prolog 知識庫的形式呈現職責劃分，每項論點均標註其主要來源。

Keywords: 巴西、估值專業能力、engenheiro de avaliações、corretor de imóveis、
NBR 14653、CONFEA/CREA、COFECI/CRECI、CNAI、專屬職業

1 一項令

感到驚訝的安排
一位首次閱讀巴西估值標準的美國估價師，通常會在「定義」部分停下腳步。在由 ABNT

（巴西國家標準化機構）頒布的 ABNT NBR 14653-1:2001 標準第 3.19 條中，對「進行估值的
人」的定義如下：

engenheiro de avaliações ：具備大學學歷的專業人士，擁有進行評估所需的法定資
格及技術與科學培訓，並已向工程、建築與農學區域委員會（CREA）完成正式註
冊。(Associação Brasileira de Normas Técnicas 2001)

估值工程師須具備大學學位，符合法律、技術及科學方面的資格要求，並已在區域工程、
建築及農學委員會註冊；此職位有別於持牌或認證的住宅估價師。該標準可追溯至 2001 年。
自第 12.378/2010 號法令起，建築師已改由獨立的建築與都市規劃委員會（CAU）負責註冊，
而聯邦工程與農學委員會（CONFEA）及各州工程與農學委員會（CREAs）則僅負責工程與農
學領域的註冊事宜 (Brasil 2010)。
巴西的主要差異並不在於執照的核發—這項做法在國際上相當普遍—而在於執照的歸屬單

位。在美國，估價師由各州依據全國統一的標準核發執照，且各州的執照核發計畫受聯邦政府
監督；經紀人則另行取得執照，且通常透過交易收取報酬。相較之下，在巴西，估值被視為一
項工程活動，其執照由與負責核發結構及公共工程設計工程師執照相同的理事會體系核發；而
經紀人則受制於不同的法律與專業體系。
本文闡述了相關的專業組織，說明各組織的角色與職責，界定了它們之間的職權範圍，探

討了相關成本，並檢視了此種安排所能解決或未能解決的問題。雖然本文並未聲稱巴西的做法
更為優越，但指出巴西的制度具有連貫性且界定明確，因此應予以審慎考量，而非僅視其為一
種地方上的特殊現象。

2 估值工程師

該保留條款並非源自該標準。其依據為法律規定，且理事會已明確指出這一點。1966 年 12
月 24 日第 5.194 號法律，該法規範工程、建築及農學之執業，其第 7 條 (c) 項將「estudos,
projetos, análises, avaliações, vistorias, perícias, pareceres e divulgação técnica」列為上述專業
之執業活動；第 8 條則規定，此類活動僅限於依法具備執業資格之自然人執行，即「(Brasil
1966)」。第 6 條 (a) 項規定，未註冊者從事這些專業保留的行為屬違法行為；該法令未列
明具體哪些行為屬於此類，而 CONFEA 第 345/1990 號決議指出，估值屬於其中之一，因
其屬於該等專業的 atribuição privativa，並宣告由他人進行的估值無效 (Conselho Federal de
Engenharia, Arquitetura e Agronomia 1990)。聯邦法規中明確將「估價」列為工程師、建築師
及農藝師的執業範疇。
該理事會決議闡明，此乃授予而非創設。NBR 14653-1第 1條的註釋記載，根據 CONFEA

第第 345 號決議，
vistorias, perícias, avaliações e arbitramentos relativos a bens móveis e imóveis, suas
partes integrantes e pertences, máquinas e instalações industriais, obras e serviços
de utilidade pública, recursos naturais e bens e direitos

之活動係 atribuição privativa ——專屬職權—— 隸屬於向 CREA 註冊之各專科工程師、建築
師、農業工程師、地質學家、地理學家及氣象學家，(Associação Brasileira de Normas Técnicas
2001; Conselho Federal de Engenharia, Arquitetura e Agronomia 1990) 。一項配套決議案，即
1973 年 6 月 29 日第 218 號決議，則明定哪些職權屬於該專業的哪些類別 (Conselho Federal
de Engenharia, Arquitetura e Agronomia 1973) 。
這兩份文件所列舉的行為並不完全相同，而這項差異值得注意。Arbitramento 一詞出現在

決議中，但未見於第 7(c) 條；pareceres（技術意見）則出現在該法令中。因此，根據巴西法
律，「意見」並非本質上屬於經紀業務的範疇：自 1966 年起，工程專業便擁有提供技術意見的
權力。區分這兩項職權的關鍵不在於「意見」一詞本身，而在於其所涉及的標的。
這些行為本身需要加以定義，因為英語中沒有單一詞彙能涵蓋其中任何一項。Vistoria 指

「查驗」：透過對事實要素的詳細檢查與描述來確立事實，而不探究其成因。Perícia 指「專
家鑑定」：釐清事件的成因，或主張權利之依據，通常是為法院進行。Avaliação 指「估價」。
Arbitramento 則是指在技術上具爭議性的選項之間，或在取決於主觀因素的選項之間作出抉
擇 (Conselho Federal de Engenharia, Arquitetura e Agronomia 1990)。這三者共同保留給同一
類專業人士。根據本期刊的解讀，它們構成了一項專業能力：即審查實體物件並提出技術性結
論，而他人將依賴此結論的能力。
由此衍生出兩項後果，而這正是巴西體制特徵的根源。
首先，這項保留權是針對「職業類別」而非「專業領域」。估值權並非由任何單一專業所獨

佔。工程師、建築師及農業工程師在法律上享有同等地位，而建築師現今則是透過加拿大建築
師協會（CAU）行使此項權利。
其次，這項工作承載著工程師的責任。該產品為 laudo de avaliação，根據 NBR 14653-2 規

範，其必須標示 qualificação legal completa e assinatura do(s) profissional(is) responsável(is)
pela avaliação——即負責專業人員（或多名負責專業人員）的完整法定資格及簽名 (Associação
Brasileira de Normas Técnicas 2011)。laudo 並非服務提供者所提出的意見，而是一份經簽署
的技術文件，其作者須向具備紀律處分權的委員會對該文件負責。

3 房地產經紀人

房地產經紀人（corretor de imóveis）的規範依據是另一項完全不同的法規：1978 年 5 月 12 日
頒布的第 6.530 號法律，該法律確立了此職業的現行體制，並規範了 COFECI（聯邦房地產經
紀人理事會）及各區域 CRECIs（聯邦與區域房地產經紀人理事會）的運作，這些機構負責該
職業的註冊與紀律管理 (Brasil 1978)。
此處有兩項條文與此相關。第 2 條允許持有「Técnico em Transações Imobiliárias」頭銜者

入境，該頭銜屬於中級技術資格。相較之下，《NBR 14653-1》第 3.19 條則要求須具備大學學
位。這種結構上的差異顯示，這兩項專業在設計上本就不同，並非等同的職業途徑。

第 3 條載明經紀人的職權範圍：

房地產經紀人得從事不動產買賣、交換及租賃之仲介業務，並得就不動產交易提出
意見。(Brasil 1978)

經紀人的職責範圍涵蓋不動產之買賣、交換及出租之中介業務。此外，經紀人亦可 就不動
產之商業化提出意見。該法規使用動詞「opinar」，其賓語為「comercialização」，賦予經紀人就
不動產之行銷方式提供意見的權限，但無權從事「avaliação」。
在英語中，這種區別往往不甚明確，因為「鑑價」（appraisal）、「估值」（valuation）、「估

價」（estimate）和「價值意見」（opinion of value）等術語的涵義有所重疊，且經紀人的價格意
見（broker’s price opinion）十分常見且普遍被接受。然而在巴西，這兩項行為則由不同的法規
分別賦予給隸屬不同專業委員會、具備不同入行門檻且具有不同法律後果的專業。兩個專業委
員會之間的爭議，關鍵在於「opinar quanto à comercialização」與「avaliar」之間的差異。

4 標準的界線何在

該準則明確闡明了此項區分，儘管這種明確性具有排他性。「corretor」一詞並未出現在巴西的
估值準則中，而「engenheiro」一詞在第一部分中出現了二十三次，「arquiteto」則出現了六次。
短語「engenheiro de avaliações」在第二部分中出現了二十次。該準則並未明確禁止經紀人；它
只是將其排除在適用範圍之外。
第 2 部分中的兩項條款闡明了該標準的要求。
第 8.2.1.4.2 條規範了在對可比物件進行逐項調整時所使用的調整係數。若已公布的係數是

由公認的技術機構，或在 CONFEA/CREA 註冊的大學或公共機構所發布，且相關研究報告由
工程或建築專業人士撰寫，則該等係數可被接受；否則，估價工程師必須自行推導該等係數，
並將推導方法及計算過程附於報告中。
前文已引述的第 10.1(m) 條規定，laudo 必須載有作者的完整法律資格及簽名。
這兩項條款共同界定了相關流程：根據工程標準及 CONFEA 的規定，巴西的估價報告必

須由註冊工程專業人員編製，採用其來源由工程體系管控的調整係數，並由一名在必要時理事
會可聯繫的人士簽署。本文並未探討該標準的定量方法論，例如基礎等級、精確度、樣本規模
或顯著性水準，因這些內容已在本期其他文章 《邁向估值工程協議》中詳述。本文的重點在於
管轄權：該標準既規定了取得估值的方法，也明確了具備執行資格的人員。
另有條款對該系統中「估值」的定義產生了重大影響。第一部分第 3.8 條將「合理範

圍」（campo de arbítrio ）定義為：

Intervalo de variação no entorno do estimador pontual adotado na avaliação, den-
tro do qual pode-se arbitrar o valor do bem, desde que justificado pela existência
de características próprias não contempladas no modelo. (Associação Brasileira de
Normas Técnicas 2001)

此處所指的範圍圍繞著點估計值——第二部分規定其上下各為百分之十五——專業人士可在
該範圍內確定價值，但須以該物業中模型未能涵蓋之特徵作為依據 (Associação Brasileira de
Normas Técnicas 2011) 。該標準針對模型無法涵蓋之處，確立了一個界定明確且具合理依據
的專業判斷範圍。將此納入國家標準實屬難得：不僅明定範圍，更要求提供合理依據，且報告
必須由合格人員簽署。

5 供應問題

這項保留條款能否發揮實效，取決於可用的專業人才數量。COFECI 以首都以外地區工程師短
缺為由，為 CNAI 途徑辯護；本刊未發現任何數據可驗證此說法。
經紀行業的規模反映了市場需求。根據 COFECI 於 2024 年 4 月進行的人口普查，目前共

有 404,316 名註冊經紀人 (Conselho Federal de Corretores de Imóveis 2026)。目前尚未取得關
於從事估價工作的 CREA 註冊專業人士人數，或其執業地點的相應統計數據。即使 COFECI

所描述的人手短缺確實存在，這也無法證明該保留條款不正確；對於任何其他持牌專業而言，
情況亦可能如此，畢竟該保留條款確實提供了實質價值的回報。然而，這將構成一項成本，且
該成本在全國各地的負擔程度不均，這正是 COFECI 提出應在該限制性體系內尋求更經濟方
式來解決此問題的主因。

6 平行曲目

為此，相關機構應運而生。COFECI 設有「經認證的經紀估價師」（avaliadores de imóveis，
簡稱經紀估價師）全國登記冊，並輔以認證制度。該制度包含「經認證的估價學校」（escolas
credenciadas de avaliação）名錄、CNAI 證書、數位印章，以及定期進行的 CNAI 人口普查
(Conselho Federal de Corretores de Imóveis 2026)。完成認證課程的經紀人將被列入名冊，並獲
頒 CNAI（全國房地產估價師名冊）的證書。該證書並未賦予任何法定權力：根據第 6.530/1978
號法律第 3 條，每位經紀人原本即有權就商業化事宜發表意見。《期刊》尚未審閱建立此流程
的理事會決議，因此未對該等經紀人所出具的文件進行識別或描述。
人口普查數據闡明了本專題的宗旨。全國範圍內，CNAI 註冊的估價師佔註冊經紀人的

9%。估價師數量最多的十個州佔總數的 60%，而另外五個州則佔 43%。然而，若考量各州經
紀人的比例，估價師最集中的地區並非位於最大市場。托坎廷斯州以百分之二十居首，其次是
阿馬帕州，佔百分之十九 (Conselho Federal de Corretores de Imóveis 2026)。
有必要釐清此培訓課程所提供與未提供的內容。該課程並未擴大經紀人的法定權限：第

6.530/1978 號法令第 3 條僅允許就商品化提出意見，而完成經認證的課程並不會改變此規定。
CNAI 提供額外的培訓、資格認證、註冊制度及特定文件。此做法究竟是解決了真正的短缺問
題，還是侵犯了保留職責，兩大理事會曾就此展開訴訟；CONFEA 於 2012 年向聯邦最高法院
提出的上訴未獲支持。

7 衝突中的市議會

該爭議於 2012 年上訴至聯邦最高法院：在案號 ARE 708.474 的案件中（針對駁回受理特別上
訴之決定提出之上訴），上訴人為聯邦工程、建築與農學聯邦理事會，被上訴人為房地產經紀
人聯邦理事會，而巴西工程評估與鑑定研究所（IBAPE）則以利害關係人身分參與訴訟。本案
係由聯邦第一地區法院（聯邦區）（原案號 2007.34.00.010591-0）上訴至聯邦最高法院。2012 年
10月 1日，負責本案的法官（relatora）卡門·盧西亞（Cármen Lúcia）駁回上訴許可申請；2012
年 10 月 16 日，第二合議庭一致裁定駁回針對其裁決提出的合議庭上訴（agravo regimental）。
1

本刊之所以報導此訴訟案，是因為當事人及結果均已明確：CONFEA 為上訴人，COFECI
為被上訴人，而 IBAPE 則為利害關係人。然而，本刊不會在已知結果之外，對法院的裁決進
行進一步的定性。欲了解法院裁決詳情者，請參閱書面判決書（acórdão；聯邦最高法院，第二
合議庭，ARE 708.474 AgR，主審大法官卡門·盧西亞，2012 年 10 月 16 日審理，DJe 220，
2012 年 11 月 8 日公布）。

8 巴西的答案是什麼，又不是什麼

人們可能會傾向於將巴西的做法視為解決市場價值相關疑慮的方案，特別是針對「在波動劇
烈的市場中，於特定日期設定的單一數值，作為十年期的擔保依據並不可靠」這項批評。然
而，正如該標準所明確指出的，事實並非如此。第一部分第 3.44 條將「公允價值」（valor de
mercado）定義為「在特定基準日（Quantia mais provável pela qual se negociaria voluntariamente
e conscientemente um bem, numa data de referência, dentro das condições do mercado vigente）

1Confirmed by consulting the Supreme Federal Court’s own process record (internal no. 4292636, por-
tal.stf.jus.br), accessed on 20 September 2026: the parties involved, the origin of the case and both of the
decisions in question. The Journal has not read the reasons given for either of the decisions and therefore
includes no comment on those grounds.

下，在現行市場條件下，該資產在特定基準日（(Associação Brasileira de Normas Técnicas
2001)）上自願且知情地進行交易時，最可能的金額」（）。此處所指的，是在參考日期當下，於
現行市場條件下，該資產最可能被自願且知情地交易之金額。此定義與其他任何國際標準一
樣，同樣是以「特定時間點」及「當前市場條件」為基礎。巴西標準並未涉及「持久性」問題；
其探討的是另一項議題。
應明確闡明這項區別，因為這兩項概念常被混淆：

• 巴西針對估價作出具體規定，明定必須由誰進行估價、應使用哪些數據、所需的佐證程
度、聲明中的判斷範圍、負責簽署人以及負責的委員會。這闡明了估價的可靠性及其依
據為何。

• 德國則引入了第二項評估基準。這包括根據《聯邦不動產估價條例》（ImmoWertV）所
規定的市場價值，以及根據《抵押貸款價值條例》（BelWertV）所規定的 Beleihungswert
（即抵押貸款價值）(Bundesrepublik Deutschland 2022)。這是一個刻意設定的保守數值，
旨在確保其在整個貸款期限內保持有效；歐盟《資本要求條例》（第 229 條，2024 年修訂
版）同樣要求採用審慎保守的標準 (European Parliament and Council 2024)。這涉及價
值的持久性問題。

這些方法是相輔相成的，而非相互競爭的。關於第二種方案的進一步資訊，請參閱本期《國
際估價標準之比較》一文，該文詳細探討了德國的「雙價值體系」。一個國家可以同時採用這兩
種體系。巴西已極為嚴格地實施了第一種體系；而在所查閱的巴西文獻中，並未發現此類第二
種、持久的價值基礎。
巴西的案例並未為估值提供更堅實的依據，而是為英語系估值界常被討論的一個議題提供

了實際例證：估值的可靠性取決於估值師（鑑價師）與估值流程，並可據此加以規範。該標準
在其定義中提及專業人士。法規對該行為設有保留條款，而理事會的決議則對該保留條款作出
了詳細闡述。報告中須包含簽名及法律資格。允許的判斷範圍以數值形式明確規定，且必須以
書面形式說明理由。
正如 COFECI 所指出的，相關問題在於該專業從業者人數過少，無法滿足全國需求，因此

發展出一套並行的資格認證體系，以因應原有體系未能滿足的需求。這兩方理事會的說法均不
完全正確；一方著重確保專業能力與問責制，另一方則致力解決其所謂「僅憑專業能力無法化
解」的人才短缺問題。這種張力並非巴西獨有；凡是將專業資格限定於特定群體之處，此類張
力便會產生，而由於巴西已正式界定其範圍，此現象在此地尤為顯著。

A 巴西術語

以下詞彙是讀者於巴西《laudo》中會接觸到的內容。標有條款編號的條目在該標準中已有定
義，並直接從該標準翻譯而來：Associação Brasileira de Normas Técnicas (2001) 為一般部分，
Associação Brasileira de Normas Técnicas (2011) 為都市不動產部分。標示為 「實務用法」的
條目雖為巴西從業人員當前普遍使用，但未在上述任一部分中定義，之所以收錄，是因為它們
出現在引用該標準的報告中——此區別值得保留，因為從業人員的慣例並不具備規範性定義的
權威性。
文中刻意未列出具體數值。基礎與精確度的等級、最小樣本規模及顯著性水準，均詳載於

本期其他篇章的 〈邁向估值工程協議〉一文中。

A.1 人物、機構與文件

engenheiro de avaliações 估價工程師。一名具備高等教育背景、擁有法律資格，且在技術
及科學方面具備進行估價能力，並已向 CREA 正式註冊之專業人士（1-3.19）。

engenharia de avaliações 估值工程。應用於資產估值的專業技術與科學知識體系（1-3.15）。
此學科概念，有別於從業者。

engenharia legal 法律工程學。工程學中處於技術與法律交界處的領域，涵蓋各類評估及專
家工作（1-3.18）。

perito, perito avaliador 法院指派的專家。這是一種職能，而非獨立的職業：指已在該領域
具備專業資格、並被指派處理特定案件的專業人士。實務用法——該標準定義了該活動
（perícia），但未定義該職位。

assistente técnico 由當事人指派之技術助理，負責審查或對 perito 所做之工作提出異議。實
務用法。

corretor de imóveis 房地產經紀人。依據第 6.530/1978 號法令設立，並於 CRECI 登記；
具備從事中介業務之資格，且有權就房地產銷售提供意見 (Brasil 1978)。該詞彙在 NBR
14653 的任何部分均未出現。

CNAI Cadastro Nacional de Avaliadores Imobiliários，COFECI 的全國經紀人暨估價師登記
名冊，附有經認證的估價學校及相關證書 (Conselho Federal de Corretores de Imóveis
2026)。

CONFEA, CREA 工程與農學聯邦理事會及其地區理事會——sistema CONFEA/CREA，
採用該標準中的表述（2-3.20）。估值工程師所隸屬的登記機構。

COFECI、CRECI 聯邦房地產經紀人理事會及其各區域理事會，依據第 6.530/1978 號法律
(Brasil 1978) 規範。

entidades técnicas reconhecidas 獲認可的技術機構：代表估價工程師且於國際估價師協會
（sistema CONFEA/CREA）註冊的組織（2-3.20）。

IBAPE 英國房地產估價與專家證據學會（Instituto Brasileiro de Avaliações e Perícias de
Engenharia）。該學會為學術組織，而非發牌機構；其名稱在第 2 部中出現了十一處。

laudo de avaliação 估價報告。由估價工程師依據相關標準編製之技術報告，用以評估資
產價值（1-3.29）。第 2 部分規定，該報告須載明負責專業人員完整的法律資格及簽名
（2-10.1(m)）。

parecer técnico 技術意見。由具備專業能力且 符合法律資格的專業人士，針對其專長領域
內的事項所發布的詳細報告或技術說明（1-3.34）。請注意，該標準本身的定義包含一項
資格要求；在此術語中，技術意見並非不受規範的陳述。

ART, RRT 分別向 CREA 及 CAU 提交的技術責任聲明，藉此由專業人士對某項工程承擔責
任。此術語見於本文所引用的資料來源；未在 NBR 14653 中定義。

A.2 價值及其基礎

avaliação de bens 資產估價。由估價工程師進行的技術分析，旨在釐清一項資產的價值、
其成本、收益及權利，並針對特定目的、情境及日期，確定其經濟利用可行性的指標
(1-3.5)。該法案的定義中明確指明了執行此項工作的專業人員。

valor de mercado 市場價值。指在參考日期當下，於現行市場條件下，一項資產可自願且知
情地進行交易時，最可能達成的金額（1-3.44）。

valor em risco 風險價值。該資產中擬投保部分的價值（1-3.45）。

valor patrimonial 資產淨值。指自然人或法人全部資產所對應的價值 (1-3.46)。

preço 價格。指一項資產、產品或權利之交易實際達成或擬達成時的金額（1-3.38）。此處與
「價值」一詞始終有所區別。

custo 成本。指為生產、維護或取得一項資產所必需的直接及間接支出總和（1-3.9），其中
custo de reprodução （重置成本，未扣除折舊，1-3.9.4），custo de reedição （考慮資產現
狀後的重置成本減去折舊，參見 1-3.9.3）以及 custo de substituição （重置成本，指具有
相同功能及類似特徵的資產，參見 1-3.9.5）。

campo de arbítrio 判斷範圍。指在估值中採用的點估計值周圍的變動區間，在此範圍內，
資產價值可進行仲裁，但須以模型（1-3.8）中未考慮到的特定特徵為依據。第 2 部分將
其寬度設定為中央趨勢估計值（2-8.2.1.5.1）兩側各百分之十五。

liquidação forçada 強制清算。指出售屬強制性質，或將在短於市場平均消化期（1-3.30）內
進行的情況。

fator de comercialização 市場性係數。指資產之市場價值與其再生產或重置成本之比率，
該比率可能大於或小於一（1-3.20）。

data de referência 參考日期——估值的生效日期，此術語在整份標準中皆有出現，包括在
市場價值的定義（1-3.44）中。

aproveitamento eficiente 有效利用。指在參考日期當下，針對該地點而言，在考量周邊實
際且有效的市場趨勢後，於法律允許的用途範圍內，既值得推薦且在技術上可行的利用
方式（2-3.1）。此為巴西對應於「最高最佳用途」的概念，其涵義較美國術語更為狹窄，
因為其中已將鄰近地區的市場趨勢納入考量。

situação paradigma 範例情境。一種用作評估資產價值參考依據的假設性情境（1-3.41）。

A.3 該物業及其狀況

imóvel 不動產。由土地及其上任何附屬設施所組成的資產；可分為都市或鄉村兩類（1-3.25）。

benfeitoria 改良。對資產進行之工作或服務所產生的結果，該資產若不加以破壞或破損則無
法移除（1-3.7），可細分為：necessária（對維護該資產不可或缺），útil（雖非必要，但可
增進或便利其使用），以及 voluptuária （僅為取樂或娛樂）（1-3.7.1–3.7.3）。

gleba urbanizável 可開發地塊。指可透過分區、分割或開發等方式，以利高效利用而建置都
市基礎建設的土地（2-3.30）。

desmembramento 土地分割。利用現有道路系統將土地劃分為建築地塊，且不開闢新的公共
道路（2-3.14）。

domínio 統治權。一種物權，使財產在法律上、絕對且排他性地服從於某人的權力與意志
（2-3.15），並參照 domínio direto、domínio útil 及 domínio pleno，以區分 enfiteuse 下的
「裸權」與「受益權」—永佃權（emphyteusis），即以固定年度地租為代價，永久使用他人
土地的權利，美國法律中並無完全對應的制度；自 2002 年《民法典》（2-3.16–3.18）頒布
以來，不得再創設此類權益。

servidão 地役權。為另一處不動產之利益而課予某處不動產之特定負擔（1-3.40）；「受役
地」（área de servidão）係指受直接影響之從屬不動產之部分（2-3.2）。

estado de conservação 保存狀態。指資產在特定時刻的物理特性狀況，此狀況係其歷經使
用與維護所致（2-3.21）。

depreciação 折舊。因下列因素導致的資產價值減損：資產狀態或品質的變化 (1-3.12)、de-
crepitude（在正常使用下的自然老化）、deterioração（使用或維護不當）、mutilação（移除
原本存在的組件）以及 obsoletismo（技術或功能上的淘汰）（1-3.12.1–3.12.4）。Depreciação
física 在第 2 部分（2-3.12）中另行定義。

idade real, idade estimada 實際與 估計年份：自建築工程實際竣工以來經過的時間，以及
根據該物業的構造、建築及功能特徵所推算出的近似年份（2-3.32、2-3.31）。

defeitos construtivos 施工瑕疵。指因設計、施工或所用材料之缺陷所導致，可能造成實際
損害或威脅使用者健康或安全之異常狀況（2-3.11）。

gabarito de altura 法律規定在特定地點允許的 最高高度（2-3.29）。

padrão construtivo 施工標準。改善工程的品質係依據設計規範、材料、工藝及實際投入的
勞務來判定（2-3.49）；實務上則參照 SINDUSCON 公布的區域 CUB 成本表。

Ross–Heidecke 一套結合使用年限與保存狀態的折舊標準，在巴西廣為教授與應用。實際應
用方面——該標準雖界定了折舊概念，但並未規定具體的折舊比例。

matrícula 位於 cartório de registro de imóveis 的物業登記卷宗，其性質更接近產權摘要，而
非評估員的地塊記錄。實務用法。

A.4 方法、數據與推論

dado de mercado 市場數據。指在市場中針對特定資產所蒐集的一組資訊（1-3.10）。

pesquisa 研究。識別、調查、蒐集、篩選、處理、分析及詮釋市場數據結果的相關活動（1-
3.36）。

amostra, população 樣本與 總體：一組能代表總體的市場數據 (1-3.2)，以及待分析市場區
隔中的全部市場數據 (1-3.37)。

tratamento de dados 數據處理。應用以相對值表示市場數據與被估價物業之間屬性差異的
運算（1-3.43）。

homogeneização 均質化。透過數學變換處理觀察到的價格，藉此以相對值表示市場數據的
屬性與待估價物業的屬性之間的差異（1-3.24）。

inferência estatística 統計推論。統計學中那部分內容，允許根據樣本對總體做出推論（1-
3.26）。

modelo, modelo de regressão 模型：對現實的技術性表徵（1-3.31）；以及用於根據樣本來
表徵某種現象的迴歸模型（1-3.32）。

nível de significância 顯著性水準。當零假設為真時，拒絕該零假設的機率（1-3.33）；而零
假設本身則定義為 1-3.23。

estimativa pontual, estimativa de tendência central 點估計值與 集中趨勢估計值（2-
3.24、2-3.23）— 用以衡量「campo de arbítrio」的數值。

códigos alocados, códigos ajustados 分配與 調整後的編碼：一種用於質性特徵的有序邏輯
量表，以及透過對二分變數進行迴歸分析，從樣本本身提取出的量表（2-3.7, 2-3.6）。

conciliação 校正。在採用多於一種方法時（2-3.8），應依據所獲得的結果，並經充分論證後，
採納最終數值。

estrutura , conduta , desempenho do mercado 市場結構、行為與表現 ——對主要參與
者的分析性分解、他們為影響交易所採取的做法，以及市場在某個期間內演變的證據
（2-3.10、2-3.9、2-3.13）；綜合上述內容，conjuntura do mercado （2-3.10）。

A.5 該報告

finalidade 預期用途 — 估值所服務之目的（出售、擔保、會計、保險、徵收等）。

objetivo 目標——估算對象為何：出售市場價值、租金價值，或其他基準。

vistoria 檢查。這是《第 5.194/1966 號法令》第 7(c) 條及 CONFEA 第 345/1990 號決議
(Brasil 1966; Conselho Federal de Engenharia, Arquitetura e Agronomia 1990) 所保留的
行為之一；該術語在第 2 部分中共出現十四次。

perícia 專家鑑定。由具備特定資格的專業人士進行的一項技術活動，旨在查明並釐清事實、
核實資產狀況、釐清事件成因，以及評估資產、其成本、收益或權利（1-3.35）。

arbitramento 價值仲裁。根據 CONFEA 第 345/1990 號決議，與其他三項法案一併予以保
留 (Conselho Federal de Engenharia, Arquitetura e Agronomia 1990)；但與其他三項法
案不同的是，該法案並未列入第 5.194/1966 號法律第 7 條 (c) 項的清單中。

grau de fundamentação, grau de precisão 一份巴西報告所提出的兩項獨立衡量標準—研
究結果的可靠性，以及估計值周邊區間的緊密度。這兩者並非同一回事；當市場波動劇
烈時，一份報告可能既具可靠性卻又缺乏精確度。相關表格請參閱本期其他頁面的 邁向
估值工程協議。

pressupostos, ressalvas e fatores limitantes 假設、限制條件與限制因素——巴西對應於
美國限制條件的對應概念。

關於參考文獻的說明。條款編號的標示方式如下：針對 ABNT NBR 14653-1:2001 採用 1-n ，
針對 ABNT NBR 14653-2:2011 則採用 2-n 。引用內容均依據 2001 年版，該版本已由 ABNT
NBR 14653-1:2019 取代；現行版本中的條款編號可能有所不同。出現次數統計均以官方葡萄
牙文文本為準。譯文由本期刊提供；若巴西機構無對應的美國機構—enfiteuse 即為最明顯的例
子—則保留葡萄牙文原詞，而非強行替換為可能引起誤解的等效詞。

B 權責劃分，作為知識基礎

B.1 為何要正式說明這一點？

本文所述的巴西制度，幾乎完全是透過二手摘要傳達給英語讀者，而這些摘要彼此之間存在分
歧 ——最常見的分歧在於將「corretor de imóveis」視為某種評估人，但巴西的法律文件並未
如此認定。2 散文式的描述同樣會導致這種誤解。但正式的陳述則不然：它可以被質疑、可以
被反駁，且其中每個命題皆可被指定為其來源的條款名稱。
下方的知識庫將權限劃分為 Prolog 事實與規則。Trealla 是本期刊所採用的 Prolog 實作，

同時亦將與 SWI-Prolog 及 Scryer Prolog 的相容性作為必要條件予以維持；因此，該知識庫採
用標準的 ISO Prolog格式，不含任何特定於單一系統的結構。prolog/tests/test_valuation_roles.pl
中的五十一項測試在上述三種系統下均能通過。

B.2 來源閘門

知識庫中的任何內容均未直接斷言。每個命題皆以

claim(Fact, SourceId, Status) Status ∈ { verified , unverified }

的形式封裝，且僅有 verified 聲明能透過 fact/1 觸及一般關係。此處的 「已驗證」僅指一
件事：即已閱讀所引用的主要文本。unverified 聲明僅對 provisional/1 及審計謂詞可見，
對其他任何內容均不可見。
在本文中，「閘門」確實發揮了實質作用。有兩項讀者可能預期會發現的論點，因相關來

源尚未被查閱，故刻意未納入分析範疇：其一，農業工程師的鄉村工作受 NBR 14653-3 規範；
其二，《民事訴訟法典》中對法院指定專家的處理方式。這兩項內容雖已連同引用來源一併記
錄在案，但均未產生實質效力。建築師依據第 12.378/2010 號法令從 CREA 系統轉入 CAU 的
情況，現已透過法規查閱並經 (Brasil 2010) 驗證：該知識庫將 CAU 記錄為註冊建築師的第三
個理事會，同時仍將建築師保留在工程理事會體系內—根據 NBR 14653-1 及第 345/1990 號決
議，該職業即歸屬於此體系。美國的整個對照集亦屬此類，其存在旨在使比較能被明確表述為
「比較」，且本附錄中的任何查詢均無法觸及該對照集。

謂詞 ‘ check/0‘ 會一併執行這些結構性檢查：每個主張皆引用已宣告的來源；每個角色皆
由某個理事會登記，並歸屬於某個理事會家族；沒有任何角色在同一項行為中具有相互矛盾的
地位；沒有任何行為未被指派；且所有項目皆附有英文標籤。

2The word corretor does not occur anywhere in either part of the valuation standard: zero occurrences in
Associação Brasileira de Normas Técnicas (2001) and zero in Associação Brasileira de Normas Técnicas (2011).

B.3 知識庫中的說明

該知識庫將職務分為兩個 理事會體系—此術語源自《期刊》，並非巴西用語—並指明專業人
士註冊於哪個聯邦理事會體系，以及受哪個體系的紀律管轄：CONFEA/CREA 體系，抑或
COFECI/CRECI 體系。3

層級至關重要。在工程師理事會體系中，多個專業共享相同的專屬職能，因此沒有任何單
一角色能完全獨佔「avaliação」的權限；而在經紀人理事會體系中，CNAI 資格認證雖增添了
一項工具，卻未擴大法定職權範圍。因此，本文所探討的區別位於個人職稱之上的一個層級，
而表達此區別的謂詞是「council_family_distinct_task/2」。

行為 工程學 經紀業務 來源

vistoria 專屬 — 第 345/1990 號決議
perícia 專屬 — 第 345/1990 號決議
avaliação 專屬 有爭議

（第 345 號決議第 3 條；
曾訴諸訴訟）

第 345/1990 號決議

arbitramento 專屬 — 第 345/1990 號決議
intermediação — 允許 1978 年第 6.530 號法律第 3 條
opinião quanto à comercialização — 允許 1978 年第 6.530 號法律第 3 條

Table 1: 依據 valuation_roles.pl 所載之職權劃分。沒有任何行為同時隸屬於兩個組織；
no_act_crosses_the_two_families 之測試即證明此點。工程理事會所屬之專業組織為 CON-
FEA 第 345/1990 號決議所列之專業團體，並已向 CREA (Conselho Federal de Engenharia,
Arquitetura e Agronomia 1990) 註冊；經紀理事會所屬之專業組織則為 corretor de imóveis 及
CNAI 經紀估價師 (Brasil 1978; Conselho Federal de Corretores de Imóveis 2026) 。

該表中有兩處不對稱之處值得特別指出。第一點在於，估值（avaliação）是房仲理事會
體系中唯一存在 爭議（而非僅是保持沉默）的專業行為—第 345/1990 號決議第 3 條明定，
未向 CREA 註冊者所進行的估值均屬無效，而此規定是否適用於房仲經紀人，曾是兩大
理事會之間爭訟的焦點。第二點在於，經紀人依據法規所獲授權的本質，是 針對商業化
（commercialization）的意見，而非估值：第 6.530/1978 號法令第 3 條賦予 corretor de imóveis
中介權，即「podendo, ainda, opinar quanto à comercialização imobiliária」(Brasil 1978)。兩
大理事會之間所有爭議，皆存在於這兩個詞語之間的鴻溝之中。
市場准入的不對稱性亦已明文規定。NBR 14653-1 第 3.19 條將「engenheiro de avaliações」

定義為「nível superior」的專業人士，須具備法定資格並於 CREA (Associação Brasileira
de Normas Técnicas 2001) 註冊；1978 年第 6.530 號法令第 2 條規定，持有「Técnico em
Transações Imobiliárias」頭銜者（此為技術資格 (Brasil 1978)）可從事房地產經紀職業。相關
測試 qualification_levels_differ 則將兩者區分開來。

B.4 來源

% SPDX-License-Identifier: AGPL-3.0-or-later
% Copyright (C) 2026 William Bert Craytor. Published by The Valuation Engineer
% Journal (Pacific Vista Net). This program is free software: you can
% redistribute it and/or modify it under the terms of the GNU Affero General
% Public License as published by the Free Software Foundation, either version 3
% of the License, or (at your option) any later version. It is distributed
% WITHOUT ANY WARRANTY, without even the implied warranty of MERCHANTABILITY or
% FITNESS FOR A PARTICULAR PURPOSE; see the COPYING file beside it or
% <https://www.gnu.org/licenses/agpl-3.0.html>.

/* valuation_roles.pl -- who may value real property in Brazil, and what each
role is actually competent to do.

3The standard itself speaks of the sistema CONFEA/CREA (Associação Brasileira de Normas Técnicas 2011,
cl. 3.20), which is the same idea under a Brazilian name.

Written for Vol. 1, Issue 2, Article 5, "The Curious Case of Brazilian
Valuation Engineers". The article's claim is that Brazil does not divide the
work the way an American reader expects: valuation is an engineering
attribution reserved to CREA-registered professionals, while the real estate
broker holds a separate, statutorily narrower competence. This file states
that division precisely enough to be queried and contradicted.

Entry points:

role(?Role) a professional role
task(?Task) a distinct professional act
performs(?Role, ?Task, ?Standing) Standing in {exclusive, permitted, excluded,

contested}
issues(?Role, ?Instrument) the document the role may sign
distinct_task(?Role, ?Task) a task no other role in the same

jurisdiction may perform at all
boundary(?TaskA, ?TaskB) two acts that are routinely confused
check run every consistency rule; true if clean

PROVENANCE IS PART OF THE DATA. Nothing is asserted directly. Every fact is
wrapped in

claim(Fact, SourceId, Status) Status in {verified, unverified}

and only a verified claim reaches the ordinary relations, through fact/1.
"verified" means the editor read the cited primary text; the sources marked
so below were read on 20 September 2026 from the official ABNT PDFs held in
docs/ and from planalto.gov.br, and on 6 October 2026 from
normativos.confea.org.br (Res. 218/1973 and 345/1990) and planalto.gov.br
(Lei 12.378/2010). An unverified claim is visible to
provisional/1 and to the audit predicates, and to nothing else -- in
particular the LaTeX appendix prints only verified rows, so a recollection
cannot reach the page by accident.

This gate is why the file is worth having. The Brazilian material circulates
in English mostly as second-hand summary, and the summaries disagree. The
only cure is to record what the primary text says and to keep everything
else quarantined until someone reads the source.

Engines: Trealla (vProlog) is the Journal's official Prolog, and
compatibility with SWI-Prolog and Scryer Prolog is maintained as a
requirement, not as a convenience. The code is therefore plain ISO Prolog
with no engine-specific constructs, and all three must pass before a change
is committed; see run-tests.sh and tests/test_valuation_roles.pl. */

:- use_module(library(lists)).

:- discontiguous(claim/3).
:- discontiguous(source/3).

/* ======================================================================
SOURCES
====================================================================== */

/* source(Id, Kind, Citation) -- Kind in {standard, statute, resolution,
council_publication}. A claim may only cite an Id declared here; check/0
enforces it. */

source(nbr_14653_1, standard,
'ABNT NBR 14653-1:2001, Avaliação de bens -- Parte 1: Procedimentos gerais (ABR 2001,

valid from 30.05.2001, incorporating Errata 1 of JUN 2001 and Errata 2 of ABR
2005)').

↪→

↪→

source(nbr_14653_2, standard,
'ABNT NBR 14653-2:2011, Avaliação de bens -- Parte 2: Imóveis urbanos').

source(lei_5194_1966, statute,
'Lei nº 5.194, de 24 de dezembro de 1966 -- regulates the exercise of the professions

of engineer, architect and agricultural engineer
(planalto.gov.br/ccivil_03/leis/l5194.htm)').

↪→

↪→

source(confea_res_218_1973, resolution,
'Resolução nº 218, de 29/06/1973, do CONFEA -- fixes the professional attributions of

the engineer, architect and agronomist by modality').↪→

source(confea_res_345_1990, resolution,
'Resolução nº 345, de 27/07/1990, do CONFEA -- engenharia de avaliações and perícias

de engenharia as activities of higher-education professionals').↪→

source(lei_6530_1978, statute,
'Lei nº 6.530, de 12 de maio de 1978 -- regulates the profession of Corretor de

Imóveis (planalto.gov.br/ccivil_03/leis/l6530.htm)').↪→

source(cofeci_site, council_publication,
'Conselho Federal de Corretores de Imóveis, www.cofeci.gov.br, pages "Avaliadores de

Imóveis", "Escolas Credenciadas de Avaliação", "Certificado CNAI" and "CENSO
CNAI", consulted 20 September 2026').

↪→

↪→

source(lei_12378_2010, statute,
'Lei nº 12.378, de 31 de dezembro de 2010 -- regulates the exercise of Architecture

and Urbanism and creates the CAU').↪→

source(nbr_14653_3, standard,
'ABNT NBR 14653-3, Avaliação de bens -- Parte 3: Imóveis rurais').

source(cpc_2015, statute,
'Lei nº 13.105, de 16 de março de 2015 (Código de Processo Civil), arts. 156 and 465

-- the court-appointed expert').↪→

source(uspap, standard,
'Uniform Standards of Professional Appraisal Practice (The Appraisal Foundation)').

source(firrea_title_xi, statute,
'Financial Institutions Reform, Recovery, and Enforcement Act of 1989, Title XI (12

U.S.C. 3331 et seq.)').↪→

/* ======================================================================
THE PROVENANCE GATE
====================================================================== */

fact(F) :- claim(F, _, verified).

provisional(F) :- claim(F, _, unverified).

cited(F, Citation) :-
claim(F, Id, _),
source(Id, _, Citation).

/* ======================================================================
ACCESSORS -- the vocabulary the article and the tests speak
====================================================================== */

jurisdiction(J) :- fact(jurisdiction(J)).
role(R) :- fact(role(R)).
belongs_to(R, J) :- fact(belongs_to(R, J)).
label(X, Lang, Text) :- fact(label(X, Lang, Text)).
council(C) :- fact(council(C)).
registers(C, R) :- fact(registers(C, R)).
entry_qualification(R, Level) :- fact(entry_qualification(R, Level)).
task(T) :- fact(task(T)).
performs(R, T, Standing) :- fact(performs(R, T, Standing)).
instrument(I) :- fact(instrument(I)).
issues(R, I) :- fact(issues(R, I)).
governed_by(R, SourceId) :- fact(governed_by(R, SourceId)).
council_family(R, F) :- fact(council_family(R, F)).

/* ======================================================================
JURISDICTIONS
====================================================================== */

claim(jurisdiction(br), nbr_14653_1, verified).
claim(jurisdiction(us), uspap, unverified).

/* ======================================================================
BRAZIL -- the councils
----------------------------------------------------------------------
Two federal council systems, each an autarquia with its own register,
discipline and jurisdiction. The article's subject is the line between
them.
====================================================================== */

claim(council(confea_crea), confea_res_345_1990, verified).
claim(label(confea_crea, en, 'Federal Council of Engineering and Agronomy and its regional

councils (CONFEA/CREA)'), lei_12378_2010, verified). % renamed by art. 64↪→

claim(council(cofeci_creci), lei_6530_1978, verified).
claim(label(cofeci_creci, en, 'Federal Council of Real Estate Brokers and its regional

councils (COFECI/CRECI)'), lei_6530_1978, verified).↪→

claim(council(cau), lei_12378_2010, verified). % art. 24
claim(label(cau, en, 'Council of Architecture and Urbanism (CAU)'), lei_12378_2010,

verified).↪→

/* ======================================================================
BRAZIL -- the roles
====================================================================== */

/* The valuation engineer. NBR 14653-1 cl. 3.19 defines the role in the
standard's own definitions: a higher-education professional, legally
qualified and technically and scientifically capable of performing
valuations, duly registered with the CREA. This single clause carries most
of the article's argument. */

claim(role(engenheiro_de_avaliacoes), nbr_14653_1, verified).
claim(belongs_to(engenheiro_de_avaliacoes, br), nbr_14653_1, verified).
claim(label(engenheiro_de_avaliacoes, pt_br, 'engenheiro de avaliações'), nbr_14653_1,

verified).↪→

claim(label(engenheiro_de_avaliacoes, en, 'valuation engineer'), nbr_14653_1, verified).
claim(registers(confea_crea, engenheiro_de_avaliacoes), nbr_14653_1, verified).
claim(entry_qualification(engenheiro_de_avaliacoes, nivel_superior), nbr_14653_1, verified).
claim(governed_by(engenheiro_de_avaliacoes, nbr_14653_1), nbr_14653_1, verified).
claim(governed_by(engenheiro_de_avaliacoes, nbr_14653_2), nbr_14653_2, verified).
claim(governed_by(engenheiro_de_avaliacoes, confea_res_218_1973), confea_res_218_1973,

verified).↪→

/* The architect. Named alongside the engineer in CONFEA Res. 345/1990 and
throughout NBR 14653-1, which occurs there six times. Registration moved
from the CREA system to the CAU under Lei 12.378/2010 (art. 24 creates the
CAU/BR and the CAUs; art. 55 moves architects registered with the CREAs to
the CAUs; art. 2(VI) keeps vistoria, pericia and avaliacao among the
architect's activities), after the standard was issued -- the standard's
text still says CREA. The statute was read on 6 October 2026, so the CAU
registration is verified. The CREA registration below is kept as what the
2001 standard states, and the architect stays in the engineering council
family, where NBR 14653-1 and Res. 345/1990 place the profession: the
council-family structure has two families, and the CAU is recorded as a
third council, not as a third family. */

claim(role(arquiteto), nbr_14653_1, verified).
claim(belongs_to(arquiteto, br), nbr_14653_1, verified).
claim(label(arquiteto, pt_br, 'arquiteto'), nbr_14653_1, verified).

claim(label(arquiteto, en, 'architect'), nbr_14653_1, verified).
claim(entry_qualification(arquiteto, nivel_superior), confea_res_345_1990, verified).
claim(registers(confea_crea, arquiteto), nbr_14653_1, verified).
claim(registers(cau, arquiteto), lei_12378_2010, verified). % arts. 24, 55
claim(governed_by(arquiteto, confea_res_218_1973), confea_res_218_1973, verified).

/* The agronomist, who carries rural property. NBR 14653-3 is the rural part;
we do not hold it, so the link to rural valuation stays unverified. */

claim(role(engenheiro_agronomo), nbr_14653_1, verified).
claim(belongs_to(engenheiro_agronomo, br), nbr_14653_1, verified).
claim(label(engenheiro_agronomo, pt_br, 'engenheiro agrônomo'), nbr_14653_1, verified).
claim(label(engenheiro_agronomo, en, 'agricultural engineer'), nbr_14653_1, verified).
claim(entry_qualification(engenheiro_agronomo, nivel_superior), confea_res_345_1990,

verified).↪→

claim(registers(confea_crea, engenheiro_agronomo), nbr_14653_1, verified).
claim(governed_by(engenheiro_agronomo, nbr_14653_3), nbr_14653_3, unverified).
claim(governed_by(engenheiro_agronomo, confea_res_218_1973), confea_res_218_1973, verified).

/* The real estate broker. Lei 6.530/1978 art. 2 admits to the profession the
holder of the title of Tecnico em Transacoes Imobiliarias -- a technical,
secondary-level qualification, not nivel superior. That asymmetry with cl.
3.19 is the structural fact, and it is why the two roles cannot simply be
merged. */

claim(role(corretor_de_imoveis), lei_6530_1978, verified).
claim(belongs_to(corretor_de_imoveis, br), lei_6530_1978, verified).
claim(label(corretor_de_imoveis, pt_br, 'corretor de imóveis'), lei_6530_1978, verified).
claim(label(corretor_de_imoveis, en, 'real estate broker'), lei_6530_1978, verified).
claim(registers(cofeci_creci, corretor_de_imoveis), lei_6530_1978, verified).
claim(entry_qualification(corretor_de_imoveis, nivel_tecnico), lei_6530_1978, verified).
claim(governed_by(corretor_de_imoveis, lei_6530_1978), lei_6530_1978, verified).

/* The broker-appraiser of the CNAI track. COFECI publishes a register of
Avaliadores de Imoveis, a list of accredited valuation schools, a CNAI
certificate and a CNAI census -- the apparatus exists and was read on the
council's own site. The resolution that created it has NOT been read; no
number is asserted here rather than guessed at. */

claim(role(corretor_avaliador_cnai), cofeci_site, verified).
claim(belongs_to(corretor_avaliador_cnai, br), cofeci_site, verified).
claim(label(corretor_avaliador_cnai, pt_br, 'corretor de imóveis avaliador (CNAI)'),

cofeci_site, verified).↪→

claim(label(corretor_avaliador_cnai, en, 'broker-appraiser on the CNAI register'),
cofeci_site, verified).↪→

claim(registers(cofeci_creci, corretor_avaliador_cnai), cofeci_site, verified).
claim(entry_qualification(corretor_avaliador_cnai, nivel_tecnico_mais_curso), cofeci_site,

verified).↪→

/* The court-appointed expert is a function, not a profession: the judge
appoints a qualified professional to it. Unverified pending a reading of
CPC arts. 156 and 465. */

claim(role(perito_judicial), cpc_2015, unverified).
claim(belongs_to(perito_judicial, br), cpc_2015, unverified).
claim(label(perito_judicial, en, 'court-appointed expert'), cpc_2015, unverified).

/* ---------------------------------------------------------------------
council_family/2 -- which of the two federal council systems a role is
registered in and disciplined by: the CONFEA/CREA system (engineering) or
the COFECI/CRECI system (brokerage). "Council family" is the Journal's
term, not a Brazilian one; NBR 14653-2 cl. 3.20 speaks of the "sistema
CONFEA/CREA", which is the same idea.

The level matters. Within the engineering council family several
professions share the same reserved acts, and within the brokerage council

family the CNAI credential adds an instrument without enlarging the
statutory competence. So no single role owns an act outright, and
distinct_task/2 at the level of a role says less than
council_family_distinct_task/2 does.
--------------------------------------------------------------------- */

claim(council_family(engenheiro_de_avaliacoes, engineering), confea_res_345_1990, verified).
claim(council_family(arquiteto, engineering), confea_res_345_1990, verified).
claim(council_family(engenheiro_agronomo, engineering), confea_res_345_1990, verified).
claim(council_family(corretor_de_imoveis, brokerage), lei_6530_1978, verified).
claim(council_family(corretor_avaliador_cnai, brokerage), cofeci_site, verified).

/* ======================================================================
THE TASKS
----------------------------------------------------------------------
CONFEA Res. 345/1990, as quoted in the note to clause 1 of NBR 14653-1,
makes four acts the ATRIBUICAO PRIVATIVA -- the exclusive attribution --
of engineers, architects, agronomists, geologists, geographers and
meteorologists registered with the CREA: vistoria, pericia, avaliacao and
arbitramento, over movable and immovable property, their integral parts
and appurtenances.

Lei 6.530/1978 art. 3 gives the broker a different competence altogether:
intermediation, "podendo, ainda, opinar quanto a comercializacao
imobiliaria". The statute grants an opinion on commercialization. It does
not grant avaliacao. Everything contested between the two councils lives
in the gap between those two words.
====================================================================== */

claim(task(vistoria), confea_res_345_1990, verified).
claim(label(vistoria, en, 'inspection: direct observation of the property to establish its

condition and characteristics'), nbr_14653_1, verified).↪→

claim(task(pericia), confea_res_345_1990, verified).
claim(label(pericia, en, 'expert examination: technical determination of disputed fact,

typically for a court'), nbr_14653_1, verified).↪→

claim(task(avaliacao), confea_res_345_1990, verified).
claim(label(avaliacao, en, 'valuation: determination of the value of an asset, its fruits and

rights'), nbr_14653_1, verified).↪→

claim(task(arbitramento), confea_res_345_1990, verified).
claim(label(arbitramento, en, 'arbitration: deciding between technically controversial

alternatives, or between alternatives that turn on subjective factors'),
confea_res_345_1990, verified). % art. 1(b)

↪→

↪→

claim(task(intermediacao), lei_6530_1978, verified).
claim(label(intermediacao, en, 'intermediation in the purchase, sale, exchange and letting of

real property'), lei_6530_1978, verified).↪→

claim(task(opiniao_comercializacao), lei_6530_1978, verified).
claim(label(opiniao_comercializacao, en, 'opinion as to real estate commercialization'),

lei_6530_1978, verified).↪→

/* ---------------------------------------------------------------------
Who performs what.
--------------------------------------------------------------------- */

/* The three core acts rest on a statute, not only on a council resolution.
Lei 5.194/1966 art. 7(c) lists "estudos, projetos, análises, AVALIAÇÕES,
VISTORIAS, PERÍCIAS, PARECERES e divulgação técnica" among the professional
activities of the engineer, architect and agronomist, and art. 8 makes the
activities of art. 7(a)-(f) the competence of natural persons legally

qualified for them; art. 6(a) makes it illegal practice for an
unregistered person to perform acts reserved to these professions, but
the statute does not list which acts those are. So the statute grants
the competence (permitted), and Res. 345/1990 states the reservation
(exclusive: "atribuição privativa"; art. 3 voids valuations by others).
Both citations are recorded, so the reader can see which authority
carries which act -- arbitramento comes from the resolution, parecer from
the statute. */

claim(performs(engenheiro_de_avaliacoes, avaliacao, permitted), lei_5194_1966, verified).
claim(performs(engenheiro_de_avaliacoes, vistoria, permitted), lei_5194_1966, verified).
claim(performs(engenheiro_de_avaliacoes, pericia, permitted), lei_5194_1966, verified).
claim(performs(arquiteto, avaliacao, permitted), lei_5194_1966, verified).
claim(performs(arquiteto, vistoria, permitted), lei_5194_1966, verified).
claim(performs(arquiteto, pericia, permitted), lei_5194_1966, verified).
claim(performs(engenheiro_agronomo, avaliacao, permitted), lei_5194_1966, verified).
claim(performs(engenheiro_agronomo, vistoria, permitted), lei_5194_1966, verified).
claim(performs(engenheiro_agronomo, pericia, permitted), lei_5194_1966, verified).

claim(task(parecer_tecnico), lei_5194_1966, verified).
claim(label(parecer_tecnico, en, 'technical opinion, as a professional activity of the

engineering professions'), lei_5194_1966, verified).↪→

claim(performs(engenheiro_de_avaliacoes, parecer_tecnico, exclusive), lei_5194_1966,
verified).↪→

claim(performs(arquiteto, parecer_tecnico, exclusive), lei_5194_1966, verified).
claim(performs(engenheiro_agronomo, parecer_tecnico, exclusive), lei_5194_1966, verified).

claim(governed_by(engenheiro_de_avaliacoes, lei_5194_1966), lei_5194_1966, verified).
claim(governed_by(arquiteto, lei_5194_1966), lei_5194_1966, verified).
claim(governed_by(engenheiro_agronomo, lei_5194_1966), lei_5194_1966, verified).

claim(performs(engenheiro_de_avaliacoes, vistoria, exclusive), confea_res_345_1990,
verified).↪→

claim(performs(engenheiro_de_avaliacoes, pericia, exclusive), confea_res_345_1990, verified).
claim(performs(engenheiro_de_avaliacoes, avaliacao, exclusive), confea_res_345_1990,

verified).↪→

claim(performs(engenheiro_de_avaliacoes, arbitramento, exclusive), confea_res_345_1990,
verified).↪→

claim(performs(arquiteto, vistoria, exclusive), confea_res_345_1990, verified).
claim(performs(arquiteto, pericia, exclusive), confea_res_345_1990, verified).
claim(performs(arquiteto, avaliacao, exclusive), confea_res_345_1990, verified).
claim(performs(arquiteto, arbitramento, exclusive), confea_res_345_1990, verified).

claim(performs(engenheiro_agronomo, vistoria, exclusive), confea_res_345_1990, verified).
claim(performs(engenheiro_agronomo, pericia, exclusive), confea_res_345_1990, verified).
claim(performs(engenheiro_agronomo, avaliacao, exclusive), confea_res_345_1990, verified).
claim(performs(engenheiro_agronomo, arbitramento, exclusive), confea_res_345_1990, verified).

/* Lei 6.530/1978 art. 3 says "Compete ao Corretor de Imoveis exercer a
intermediacao": a competence, not a reservation -- it does not say
"privativa". Intermediation is therefore `permitted', not `exclusive'.

The broker's standing on avaliacao is `contested'. Res. 345/1990 art. 3
declares valuations by persons not registered with a CREA "nulas de pleno
direito"; whether that reaches the broker was litigated between the two
councils, and CONFEA's appeal to the Supreme Federal Court (ARE 708.474)
failed in 2012. The lower-court judgment has not been read, so the KB
records the dispute rather than a winner. */

claim(performs(corretor_de_imoveis, intermediacao, permitted), lei_6530_1978, verified).
claim(performs(corretor_de_imoveis, opiniao_comercializacao, permitted), lei_6530_1978,

verified).↪→

claim(performs(corretor_de_imoveis, avaliacao, contested), confea_res_345_1990, verified).

claim(performs(corretor_avaliador_cnai, intermediacao, permitted), lei_6530_1978, verified).
claim(performs(corretor_avaliador_cnai, opiniao_comercializacao, permitted), cofeci_site,

verified).↪→

claim(performs(corretor_avaliador_cnai, avaliacao, contested), confea_res_345_1990,
verified).↪→

/* ---------------------------------------------------------------------
The documents. The instrument is where the boundary becomes visible to a
reader: the laudo is a signed engineering document with a legally
answerable author, and NBR 14653-2 cl. 10.1(m) requires it to carry the
"qualificacao legal completa e assinatura do(s) profissional(is)
responsavel(is) pela avaliacao". The word "corretor" appears nowhere in
either part of the standard -- 0 occurrences in Part 1, 0 in Part 2.
--------------------------------------------------------------------- */

claim(instrument(laudo_de_avaliacao), nbr_14653_1, verified).
claim(label(laudo_de_avaliacao, en, 'valuation report, signed under the author''s full legal

qualification'), nbr_14653_2, verified).↪→

claim(issues(engenheiro_de_avaliacoes, laudo_de_avaliacao), nbr_14653_2, verified).
claim(issues(arquiteto, laudo_de_avaliacao), nbr_14653_2, verified).
claim(issues(engenheiro_agronomo, laudo_de_avaliacao), nbr_14653_2, verified).

/* CORRECTION, 20 September 2026. An earlier revision of this file recorded
an instrument `ptam' as VERIFIED against cofeci_site. That was wrong: the
council page actually read names the register ("Avaliadores de Imóveis"),
the accredited schools, the "Certificado CNAI", the appraiser's area and
the CNAI census -- and nowhere uses the term. The credential below is what
the page supports. The name and form of the work product the CNAI broker
issues have NOT been read and are therefore not asserted at all. */

claim(instrument(certificado_cnai), cofeci_site, verified).
claim(label(certificado_cnai, en, 'CNAI certificate: the credential held by a broker on the

appraisers'' register'), cofeci_site, verified).↪→

claim(issues(corretor_avaliador_cnai, certificado_cnai), cofeci_site, verified).

/* ======================================================================
THE UNITED STATES -- a contrast set, entirely unverified
----------------------------------------------------------------------
Present so that distinct_task/2 has something to be distinct FROM, and so
the article's opening claim (that an American reader finds the Brazilian
arrangement strange) can be stated as a comparison rather than an
assertion. Every row is unverified and therefore invisible to role/1,
performs/3 and the printed appendix until someone reads Title XI and the
AQB criteria. Do not cite any of it.
====================================================================== */

claim(role(certified_general_appraiser), firrea_title_xi, unverified).
claim(belongs_to(certified_general_appraiser, us), firrea_title_xi, unverified).
claim(label(certified_general_appraiser, en, 'certified general appraiser'), firrea_title_xi,

unverified).↪→

claim(entry_qualification(certified_general_appraiser, state_certification), firrea_title_xi,
unverified).↪→

claim(governed_by(certified_general_appraiser, uspap), uspap, unverified).

claim(role(us_real_estate_broker), firrea_title_xi, unverified).
claim(belongs_to(us_real_estate_broker, us), firrea_title_xi, unverified).
claim(label(us_real_estate_broker, en, 'real estate broker'), firrea_title_xi, unverified).

claim(task(us_appraisal), uspap, unverified).
claim(task(broker_price_opinion), firrea_title_xi, unverified).
claim(performs(certified_general_appraiser, us_appraisal, exclusive), firrea_title_xi,

unverified).↪→

claim(performs(us_real_estate_broker, broker_price_opinion, permitted), firrea_title_xi,
unverified).↪→

claim(performs(us_real_estate_broker, us_appraisal, excluded), firrea_title_xi, unverified).

/* ======================================================================
DERIVED RELATIONS
====================================================================== */

/* A role may actually do a task if it holds it exclusively or is permitted
it. "exclusive" is a statement about the reservation, not about
solitude: Res. 345/1990 reserves the four acts to a named class of
professions jointly, so several roles may share an exclusive attribution.
What the reservation excludes is everyone outside that class. */

may(R, T) :- performs(R, T, exclusive).
may(R, T) :- performs(R, T, permitted).

reserved(T) :- performs(_, T, exclusive).

/* distinct_task(Role, Task) -- a task this role may perform that no other
role in the same jurisdiction may perform at all. This is the predicate
the article's title asks about. */

distinct_task(R, T) :-
may(R, T),
belongs_to(R, J),
\+ ( may(Other, T),

Other \== R,
belongs_to(Other, J) ).

/* council_family_distinct_task(CouncilFamily, Task) -- an act that one council system holds
and the other cannot touch at all. This is the article's real claim.

The setof/3 is load-bearing. Enumerating council_family(R, F), may(R, T) directly
yields one solution per ROLE, so a four-act reservation shared by three
engineering professions comes back twelve times. setof/3 collapses the
pairs first, so the predicate is one solution per council family and act. */

council_family_distinct_task(F, T) :-
setof(Fam-Act, R^(council_family(R, Fam), may(R, Act)), Pairs),
member(F-T, Pairs),
\+ ( may(Other, T),

council_family(Other, G),
G \== F ).

/* shares_task(A, B, T) -- two roles holding the same act. */
shares_task(A, B, T) :-

may(A, T),
may(B, T),
A @< B.

/* barred(Role, Task) -- the role is affirmatively shut out of the task. */
barred(R, T) :- performs(R, T, excluded).

/* contested(Role, Task) -- the role's standing on the task is in dispute
between the councils: neither held nor settled as barred. */

contested(R, T) :- performs(R, T, contested).

/* boundary(TaskA, TaskB) -- two acts routinely conflated in the secondary
literature, which the Journal keeps apart. */

boundary(avaliacao, opiniao_comercializacao).
boundary(avaliacao, intermediacao).
boundary(pericia, vistoria).

/* confusable(RoleA, RoleB) -- two roles, on opposite sides of a boundary,
whose outputs a lay reader mistakes for one another: B is barred from A's
act, or its standing on it is contested. */

confusable(A, B) :-

boundary(TA, TB),
may(A, TA),
may(B, TB),
( barred(B, TA) ; contested(B, TA) ).

/* ======================================================================
AUDIT
====================================================================== */

unverified_claim(F, Id) :- claim(F, Id, unverified).

undeclared_source(Id) :-
claim(_, Id, _),
\+ source(Id, _, _).

/* A role that no council registers is either an error or a bare function
(perito_judicial is the latter, and is unverified, so it does not reach
role/1). */

unregistered_role(R) :-
role(R),
\+ registers(_, R).

/* A role holding both an exclusive and an excluded standing on one task. */
contradictory_standing(R, T) :-

performs(R, T, exclusive),
performs(R, T, excluded).

/* A task nobody may perform. */
orphan_task(T) :-

task(T),
\+ may(_, T),
\+ barred(_, T),
\+ contested(_, T).

/* A role in no council family. */
role_without_council_family(R) :-

role(R),
\+ council_family(R, _).

/* A role with no label in English. */
unlabelled(X) :-

( role(X) ; task(X) ; instrument(X) ; council(X) ),
\+ label(X, en, _).

/* check/0 -- true when the knowledge base is internally consistent. Written
to be used from a build or a commit hook, and reported on by
tests/test_valuation_roles.pl. */

check :-
\+ undeclared_source(_),
\+ unregistered_role(_),
\+ contradictory_standing(_, _),
\+ role_without_council_family(_),
\+ orphan_task(_),
\+ unlabelled(_).

/* dump/0 -- human-readable dump, for checking the file by eye. */
dump :-

write('-- roles (verified only) --'), nl,
( role(R),

label(R, en, L),
write(' '), write(R), write(' '), write(L), nl,
fail

; true ),

write('-- acts each council family holds alone --'), nl,
( council_family_distinct_task(F, T2),

write(' '), write(F), write(' -> '), write(T2), nl,
fail

; true ),
write('-- acts no single role holds alone (shared within a council family) --'), nl,
( setof(A, R3^(may(R3, A), \+ distinct_task(_, A)), Shared),

member(T3, Shared),
write(' '), write(T3), nl,
fail

; true ),
write('-- unverified claims --'), nl,
( unverified_claim(F, Id),

write(' '), write(Id), write(' '), write(F), nl,
fail

; true ).

